

Poverty Lines and Poverty Rates

Goal

In order to get any measure of poverty, it is essential to make some assumptions concerning the criteria based on which to define poverty. The approach used by LIS (and most commonly adopted in the literature), is that of creating a relative poverty line based on the level and distribution of household disposable (equivalised) income in the total population. Households are classified as poor or non-poor on the basis of whether their income is lower or higher than the relative poverty line.

Once poor households are identified, you can create an indicator to help identify the proportion of poor households (or individuals) and to measure the level of poverty. The choice of indicator used will mainly depend on the purpose of the research. In this exercise, we will calculate the main indicator of poverty incidence, the head count ratio, and the income gap ratio (an important indicator of poverty intensity).

Activity

Using the 2000 Finnish data, run the same data cleaning procedures and create the equivalence scale introduced in the previous exercise. Define the poverty line as 50% of the median equivalised income. Calculate the head count ratio (defined as the percentage of individuals living in poor households) and the income gap ratio (as explained in the guidelines).

Guidelines

- When creating a variable whose value is the same for all observations (e.g. median equivalised income or poverty line), use the **CALL SYMPUT** routine. (i) First calculate the value you expect to assign to your variable, (ii) then within a **_NULL_** data step, assign this value to a global macro variable automatically generated by the routine and (iii) finally, use this macro variable as it would have been defined via a **%LET** statement.

```
PROC MEANS ...;
  VAR ... ;
  WEIGHT ...;
  OUTPUT OUT=temp MEDIAN=x;
RUN ;
DATA _NULL_;
  SET temp;
  CALL SYMPUT("y", x);
RUN;
DATA ...;
  SET ...;
  ... ;
  myvar = &y ;
...
RUN;
```

- Again, be careful when choosing weights: use **hweight** if you want to measure household poverty, and **hweight*d4** if you are interested in individual poverty.

- The Head Count Ratio (HCR) is the percentage of poor individuals in the total population. When you create a dummy variable indicating that an individual is poor (**poor** = 0 or =1), then the mean of the indicator variable (properly weighted) will be the percentage of poor individuals.
- The Income Gap is the difference between income and the poverty line. The Income Gap Ratio (IGR) is the average income gap as a percentage of the poverty line.
- To compute the number of households considered poor in the population, use the option **SUMWGT** of the **PROC MEANS** statement.

Program

```
OPTIONS NOSOURCE NONOTES NOFMterr NODATE NOCENTER LABEL NONUMBER LS=200
PS=MAX ;
```

```
DATA pov      (DROP=hweight d4 dpi);
  SET &fi00h (KEEP=hweight d4 dpi);
  IF dpi in (. 0) THEN DELETE;
  ey = dpi /SQRT(d4);
  wt = hweight * d4 ;
RUN;
PROC MEANS DATA=pov NOPRINT;
  VAR ey;
  WEIGHT wt ;
  OUTPUT OUT=temp MEDIAN=medey;
RUN ;
DATA _NULL_;
  SET temp;
  CALL SYMPUT("m",medey);
RUN;
DATA pov (KEEP=eymed povlin gap poor wt);
  SET pov;
  poor = 0 ;
  eymed = &m ;
  povlin = &m * 0.5 ;
  IF ey < povlin THEN
  DO;
    gap = povlin-ey;
    poor = 1 ;
  END;
RUN ;
PROC MEANS DATA=pov N MEAN SUMWGT;
  VAR poor eymed povlin gap;
  WEIGHT wt;
RUN ;
```

Results

Median equivalised income	96 332
Poverty line	$96,332 / 2 = 48,166$
How many poor households are in the sample?	588
How many poor individuals are there in the total population?	277 263
Head Count Ratio	5.43%
What is the average income gap among poor individuals?	12 317
Income Gap Ratio	$12,317 / 48,166 = 25.57\%$

Comments

- The head count ratio (HCR) measures poverty incidence (i.e., the number or proportion of poor people), but gives every person equal weight no matter how far they fall from the poverty line.
- The Income Gap Ratio (IGR) measures poverty intensity or depth (how poor are the poor), but one poor person with an income of an amount x counts the same as two poor people each with an income of $x/2$. That is, the IGR measures the average income gap, but not its distribution among the poor).
- Only two indicators of poverty are mentioned here, but there are several others. These include, among the most common, the whole family of Foster-Greer-Thorbecke indicators (of which the HCR is only one), the Sen index, the Takayama index, the Clark index, and the Thon index. It is important to note that a country may score better in comparison to a second country when using a particular index, but could score worse if another index was used instead.